
SUBR1.s
/**********************************************************************/
/* COPYRIGHT IBM CORP. 2011, 2013                                     */
/**********************************************************************/
SUBR1    CSECT
SUBR1    AMODE 31
SUBR1    RMODE ANY
         SAVE  (14,12),,*          SAVE CALLER'S REGISTERS IN CALLER-
*                                  PROVIDED R13 SAVE AREA
*
         LR    r3,r0               Save usable ENVblock
         USING ENVBLOCK,r3
         LR    r4,r1               Save EFPL
         USING EFPL,r4
*
         LR    r12,r15             SET UP PROGRAM BASE REGISTER
         USING SUBR1,r12
*
         LA    r2,72+MYAREASZ      Space for savearea + my autodata
*
         GETMAIN RU,LV=(r2)        GET MY REENTRANT SAVEAREA+
         ST    r13,4(,r1)          SAVE CALLER'S SAVEAREA ADDRESS IN MY
*                                  SAVEAREA (BACKWARD CHAIN)
         ST    r1,8(,r13)          SAVE MY SAVEAREA ADDRESS IN CALLER'S
*                                  SAVEAREA (FORWARD CHAIN)
         LR    r13,r1              PUT MY SAVEAREA ADDRESS IN R13
*   END OF ENTRY CODE, BEGIN PROGRAM CODE HERE
**
         LA    r11,72(r13)         make autodata addressable
         USING MYAREA,r11
         ST    r2,mysize           remember total size for later...
*
         WTO   'SUBR1 before',ROUTCDE=11
*
         L     r5,EFPLEVAL         pointer to address of...
         L     r5,0(r5)            .. evalblock
         USING EVALBLOCK,r5
*
         L     r6,ENVBLOCK_PARMBLOCK
         USING PARMBLOCK,r6
         L     r6,PARMBLOCK_SUBCOMTB
         USING SUBCOMTB_HEADER,r6
*
         MVC   EVALBLOCK_EVLEN,=F'24'
         MVC   EVALBLOCK_EVDATA(4),EVALBLOCK_EVSIZE
         MVC   EVALBLOCK_EVDATA+4(20),SUBCOMTB_HEADER
*
         WTO   'SUBR1 after',ROUTCDE=11
*
RET      LHI   r15,0
         ST    r15,myret
         J     XIT
RET2     LHI   r15,-1              return w/ -1
         ST    r15,myret
         J     XIT
**
*   BEGIN EXIT CODE
XIT      LR    r1,r13              COPY MY SAVEAREA ADDRESS
         L     r2,mysize
         L     r3,myret
         L     r13,4(,r13)         RESTORE CALLER'S SAVEAREA ADDRESS
         FREEMAIN RU,A=(1),LV=(r2) FREE MY REENTRANT SAVEAREA+
         LR    r15,r3              SET RETURN CODE
         L     r14,12(,r13)        RESTORE CALLER'S R14
         LM    r2,r12,28(r13)      RESTORE CALLER'S R2-R12
         BR    r14                 RETURN TO CALLER
**
         YREGS
**
envb0    IRXENVB 
efpl1    IRXEFPL 
evalb    IRXEVALB
parmb    IRXPARMB
subcomt  IRXSUBCT
**
MYAREA   DSECT
*
myret    DS   F
mysize   DS   F
**
* 
MYAREASZ EQU  *-MYAREA 
         END 




asmpipc.s
***********************************************************************
* COPYRIGHT IBM CORP. 2011, 2013                                      *
***********************************************************************
*
         GBLA     &DEBUG
&DEBUG   SETA (',&SYSPARM,' INDEX ',DEBUG,')
*
ASMPIPC  CSECT
ASMPIPC  AMODE 31
ASMPIPC  RMODE ANY
         STM   R14,R12,12(R13)    Save caller's registers
         LR    R12,R15            Get base address
         USING ASMPIPC,R12        Identify base register
         ST    R13,SAVE+4         Back-chain the save area
         LA    R15,SAVE           Get addr of this routine's save area
         ST    R15,8(R13)         Forward-chain in caller's save area
         LR    R13,R15            R13 -> save area of this routine
*
         LR    R3,R1               Save EFPL
         USING EFPL,R3
*
         AIF (&DEBUG EQ 0).NDBG001
         WTO   'ASMPIPC : Entering ASMPIPC...',ROUTCDE=11
.NDBG001 ANOP
*
* Load LE CEEPIPI service routine into main storage.
*
         L     R1,efplarg
         USING argtable_entry,R1
*  Arg 1
         LHI   R3,0               1st arg index
**BLL    L     R2,argtable_argstring_ptr(0*argtelen) 1st arg ptr
         L     R2,argtable_argstring_ptr(R3)  1st arg ptr
         L     R2,0(R2)           1st arg (we know len is 4)
         ST    R2,PPRTNPTR        Save the addr of CEEPIPI routine
*  Arg 2
         AHI   R3,argtelen        2nd arg index
**BLL    L     R2,argtable_argstring_ptr(1*argtelen) 2nd arg ptr
         L     R2,argtable_argstring_ptr(R3)  2nd arg ptr
         L     R2,0(R2)           2nd arg (we know len is 4)
         ST    R2,TOKEN           Save the TOKEN
*
         AHI   R3,argtelen         3rd arg index -- for callee
         AR    R1,R3                . we'll
         ST    R1,PARMLIS@          ...pass
         LA    R1,PARMLIS@         .... only
         ST    R1,PARMPTR           ..... strings
         DROP  R1
*
* Call the subroutine, which was loaded by LE
*
CSUB     EQU   *
         AIF (&DEBUG EQ 0).NDBG002
         WTO   'ASMPIPC : about call to (CALL_SUB)',ROUTCDE=11
.NDBG002 ANOP
         L     R15,PPRTNPTR       Get address of CEEPIPI routine
         CALL  (15),(CALLSUB,PTBINDEX,TOKEN,PARMPTR,                   X
               SUBRETC,SUBRSNC,SUBFBC)   Invoke CEEPIPI routine
*                                 Check return code:
         LTR   R2,R15               Is R15 = zero?
         BZ    DONE                 Yes (success).. go to next section
*                                   No (failure).. issue message & quit
         AIF (&DEBUG EQ 0).NDBG003
         WTO   'ASMPIPC : call to (CALL_SUB) failed',ROUTCDE=11
.NDBG003 ANOP
         ABEND (R2),DUMP          Abend with bad RC and dump memory
*
* Standard exit code.
*
DONE     EQU   *
         AIF (&DEBUG EQ 0).NDBG004
         WTO   'ASMPIPC : back from call to (CALL_SUB)',ROUTCDE=11
         WTO   'ASMPIPC : Leaving ASMPIPC...',ROUTCDE=11
.NDBG004 ANOP
         LA    R15,0                Passed return code for system
         L     R13,SAVE+4           Get address of caller's save area
         L     R14,12(R13)          Reload caller's register 14
         LM    R0,R12,20(R13)       Reload caller's registers 0-12
         BR    R14                  Branch back to caller
*
* =====================================================================
* CONSTANTS and SAVE AREA.
* =====================================================================
SAVE     DC    18F'0'
PPRTNPTR DS    A                  Save the address of CEEPIPI routine
PARMLIS@ DS    A                   @PARMLIST
*
* Parameters passed to a (INIT_SUB) call.
*
INITSUB  DC    F'3'               Function code to initialize for subr
@CEXPTBL DC    A(PPTBL)           Address of Preinitialization Table
@SRVRTNS DC    A(0)               Addr of service-rtns vector, 0 = none
RUNTMOPT DC    CL255' '           Fixed length string of runtime optns
TOKEN    DS    F                  Unique value returned (output)
*
* Parameters passed to a (CALL_SUB) call.
*
CALLSUB  DC    F'4'               Function code to call subroutine
PTBINDEX DC    F'0'               The row number of Preinitialization 
*                                 Table entry
PARMPTR  DC    A(0)               Pointer to @PARMLIST or zero if none
SUBRETC  DS    F                  Subroutine return code (output)
SUBRSNC  DS    F                  Subroutine reason code (output)
SUBFBC   DS    3F                 Subroutine feedback token (output)
*

* Parameters passed to a (TERM) call.
*
TERM     DC    F'5'               Function code to terminate
ENV_RC   DS    F                  Environment return code (output)
*
* =====================================================================
* Preinitialization Table.
* =====================================================================
PPTBL    CEEXPIT  ,               Preinitialization Table with index
         CEEXPITY HLLPIPI,0       0 = dynamically loaded routine
*
         CEEXPITS ,               End of PreInit table
*
*
         LTORG
**
efpl1    IRXEFPL
argtb    IRXARGTB
argtelen EQU    ARGTABLE_NEXT-ARGTABLE_ARGSTRING_PTR
*
         YREGS
*
         END   ASMPIPC




asmpipi.s
***********************************************************************
* COPYRIGHT IBM CORP. 2011, 2013                                      *
***********************************************************************
*                                                                     *
* ASMPIPI                                                             *
*                                                                     *
*   Must be called from a REXX driver                                 *
*                                                                     *
*   1 - Load and Call CEEPIPI to initialize a subroutine environment. *
*                                                                     *
*   2 - Call IRXEXCOM to set REXX variables to return the             *
*       address of CEEPIPI and the address of the environment handle. *
*                                                                     *
* Note: ASMPIPI is NOT reentrant.                                     *
* Note: Assemble with SYSPARM(DEBUG) to enable debugging messages     *
*                                                                     *
* 2012/03/15 - Barry Lichtenstein - Clean up                          *
* 2011/02/25 - Barry Lichtenstein - Original                          *
*                                                                     *
***********************************************************************
*
* +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
         GBLA     &DEBUG
&DEBUG   SETA (',&SYSPARM,' INDEX ',DEBUG,')
* +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
*
* =====================================================================
* Standard program entry conventions.
* =====================================================================
ASMPIPI  CSECT
ASMPIPI  AMODE 31
ASMPIPI  RMODE ANY
         STM   R14,R12,12(R13)    Save caller's registers
         LR    R12,R15            Get base address
         USING ASMPIPI,R12        Identify base register
         ST    R13,SAVE+4         Back-chain the save area
         LA    R15,SAVE           Get addr of this routine's save area
         ST    R15,8(R13)         Forward-chain in caller's save area
         LR    R13,R15            R13 -> save area of this routine
*
         LR    R3,R0               Save usable ENVblock
         USING ENVBLOCK,R3
*
* *********************************************************************
* Load LE CEEPIPI service routine into main storage.
* *********************************************************************
         LOAD  EP=CEEPIPI         Load CEEPIPI routine dynamically
         ST    R0,PPRTNPTR        Save the addr of CEEPIPI routine
*
* =====================================================================
* Initialize an LE Preinitialization subroutine environment.
* =====================================================================
INIT_ENV EQU   *
         AIF (&DEBUG EQ 0).NDBG001
         WTO   'ASMPIPI : about to (INIT_SUB)',ROUTCDE=11
.NDBG001 ANOP    
         LA    R5,PPTBL           Get address of Preinitialization Tabl
         ST    R5,@CEXPTBL        Ceexptbl-addr->Preinitialization Tabl
         L     R15,PPRTNPTR       Get address of CEEPIPI routine
*                                 Invoke CEEPIPI routine
         CALL  (15),(INITSUB,@CEXPTBL,@SRVRTNS,RUNTMOPT,TOKEN)
*                                 Check return code:
         LTR   R2,R15               Is R15 = zero?
         BZ    SETVAR               Yes (success).. go to next section
*                                   No (failure).. issue message, abend
         WTO   'ASMPIPI : INIT_SUB failure RC is not 8.',ROUTCDE=11
         ABEND (R2),DUMP          Abend with bad RC and dump memory
*
* =====================================================================
* Set the REXX variables PIPIADDR and PIPITOKN for access
* to the REXX driver, for use on subsequent calls into CEEPIPI
* =====================================================================
SETVAR   EQU   *
         L     R0,TOKEN
         ST    R0,TOKENCPY
*
* *********************************************************************
* First save the PIPI information needed later,
* in order to return in REXX vars
* *********************************************************************
         MVC    vvar1(4),PPRTNPTR
         MVC    vvar2(4),TOKENCPY
*
* +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
         AIF (&DEBUG EQ 0).NDBG003
         WTO   'ASMPIPI : about to ASMPIPC',ROUTCDE=11
         WTO   'ASMPIPI about to SETVAR',ROUTCDE=11
.NDBG003 ANOP
* +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
*
* *********************************************************************
* Create a shared variable request block for last variable
* *********************************************************************
         LA    R6,shvb2
shvr2    USING SHVBLOCK,r6
         MVC   shvr2.SHVNEXT,=F'0'
         MVC   shvr2.SHVUSER,=F'0'
         MVI   shvr2.SHVCODE,SHVSTORE
         MVC   shvr2.SHVNAMA,=A(var2)
         MVC   shvr2.SHVNAML,=A(evar2-var2)
         MVC   shvr2.SHVVALA,=A(vvar2)
         MVC   shvr2.SHVVALL,=A(evvar2-vvar2)
*
* *********************************************************************
* Chaing & Create a shared variable request block for next variable
* *********************************************************************
         LR    R0,R6
         LA    R6,shvb1
shvr1    USING SHVBLOCK,R6
         ST    R0,shvr1.SHVNEXT
         MVC   shvr1.SHVUSER,=F'0'
         MVI   shvr1.SHVCODE,SHVSTORE
         MVC   shvr1.SHVNAMA,=A(var1)
         MVC   shvr1.SHVNAML,=A(evar1-var1)
         MVC   shvr1.SHVVALA,=A(vvar1)
         MVC   shvr1.SHVVALL,=A(evvar1-vvar1)
* ******
* *********************************************************************
         LA    R5,=CL8'IRXEXCOM'
         ST    R5,parm1
* ******
         LHI   R5,0
         ST    R5,parm2
         ST    R5,parm3
* ******
         ST    R6,parm4
* ******
         OI    parm4,X'80'
* ******
         LR    R0,R3               restore ENVBLOCK for the call!
         LA    R1,plist
*
* *********************************************************************
         LINK  EP=IRXEXCOM
         ST    R15,myret
*
* +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
         AIF (&DEBUG EQ 0).NDBG004
         WTO   'ASMPIPI called   SETVAR',ROUTCDE=11
.NDBG004 ANOP
* +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
*
* =====================================================================
* Standard exit code.
* =====================================================================
DONE     EQU   *
*        LA    R15,0                Passed return code for system
         L     R15,myret
         L     R13,SAVE+4           Get address of caller's save area
         L     R14,12(R13)          Reload caller's register 14
         LM    R0,R12,20(R13)       Reload caller's registers 0-12
         BR    R14                  Branch back to caller
*
* =====================================================================
* CONSTANTS and SAVE AREA.
* =====================================================================
SAVE     DC    18F'0'
PPRTNPTR DS    A                  Save the address of CEEPIPI routine
TOKENCPY DS    F                  Save a copy of the TOKEN
*
* =====================================================================
* Parameters passed to a (INIT_SUB) call.
* =====================================================================
INITSUB  DC    F'3'               Function code to initialize for subr
@CEXPTBL DC    A(PPTBL)           Address of Preinitialization Table
@SRVRTNS DC    A(0)               Addr of service-rtns vector, 0 = none
* ---------------------------------------------------------------------
* why don't we get stdout output from the HLL? we do get stderr !
* ---------------------------------------------------------------------
RUNTMOPT DC    CL255''            Fixed length string of runtime optns
         ORG RUNTMOPT
         DC    C'RPTOPTS(ON) '      
         DC    C'MSGFILE(SYSPRINT) '
*        DC    C'ENVAR(_CEE_ENVFILE_S='
*        DC      C'/u/barryl/binder/SHARE/SHARE118/PIPI/envfile )'
         ORG RUNTMOPT+L'RUNTMOPT
TOKEN    DS    F                  Unique value returned (output)
*
* =====================================================================
* Parameters passed to a (CALL_SUB) call.
* =====================================================================
CALLSUB  DC    F'4'               Function code to call subroutine
PTBINDEX DC    F'0'               The row number of Preinitialization 
*                                 Table entry
PARMPTR  DC    A(0)               Pointer to @PARMLIST or zero if none
SUBRETC  DS    F                  Subroutine return code (output)
SUBRSNC  DS    F                  Subroutine reason code (output)
SUBFBC   DS    3F                 Subroutine feedback token (output)
*
* =====================================================================
* Parameters passed to a (TERM) call.
* =====================================================================
TERM     DC    F'5'               Function code to terminate
ENV_RC   DS    F                  Environment return code (output)
*
* =====================================================================
* Preinitialization Table.
* =====================================================================
PPTBL    CEEXPIT  ,               Preinitialization Table with index
         CEEXPITY HLLPIPI2,0      0 = dynamically loaded routine
*
         CEEXPITS ,               End of PreInit table
*
* =====================================================================
* This should all be AUTODATA!
* =====================================================================
MYAREA   EQU   *
*
* =====================================================================
* REXX stuff
* =====================================================================
*
* *********************************************************************
* Shared Variable 1 name and value
* *********************************************************************
var1     DC    C'PIPIADDR'         
evar1    EQU   *
* ******
vvar1    DC    CL4'addr'         will be overwritten
evvar1   EQU   *
*
* *********************************************************************
* Shared Variable 2 name and value
* *********************************************************************
var2     DC    C'PIPITOKN'         
evar2    EQU   *
* ******
vvar2    DC    CL4'tokn'         will be overwritten
evvar2   EQU   *
*
* *********************************************************************
* Parameters for IRXEXCOM call
* *********************************************************************
myret    DS    F
mysize   DS    F
**
plist    DS    0D
parm1    DS    F
parm2    DS    F
parm3    DS    F
parm4    DS    F
*
* *********************************************************************
* Shared variable block mappings
* *********************************************************************
shvb1    ORG *+SHVBLEN
shvb2    ORG *+SHVBLEN
*
* *********************************************************************
MYAREASZ EQU  *-MYAREA
*
* *********************************************************************
*
         LTORG
*
* =====================================================================
* Control block macro mappings
* =====================================================================
envb0    IRXENVB
evalb    IRXEVALB
parmb    IRXPARMB
shrvar   IRXSHVB
*
* *********************************************************************
* Map register names to variables
* *********************************************************************
         YREGS
*
* *********************************************************************
         END   ASMPIPI




edcpipim2.c
/**********************************************************************/
/* COPYRIGHT IBM CORP. 2011, 2013                                     */
/**********************************************************************/

#include <stdio.h>
#include <stdlib.h>
#include <string.h>
#include <time.h>
#include <errno.h>

#include <irxefpl.h>
#include <irxargtb.h>
#include <irxshvb.h>
/* #include <irxevalb.h> */
#include <irxenvb.h>      


typedef int (IRXEXCOM)(char func[8], void *zero2, void *zero3, struct shvblock shvbp, int envbp0, int retcode);
#pragma linkage(IRXEXCOM, OS)

char ** environ;

int cnt = 0, envbp0=0;
main (int argc, char **argv)
{

 struct efpl * EFPLP = (struct irxefpl *) __osplist;
 struct argtable_entry * ARGTEP = EFPLP->efplarg; 
 /* struct evalblock * EVALBP = EFPLP->efpleval; */

 char varname1[] = "LEREXX";
 char varvalue1[] = "perfect together";

 IRXEXCOM* excomfunc;
 int retcode;

 struct shvblock SHVAR1 = { 0 };

 SHVAR1.shvcodes._shvcode = shvstore;

 SHVAR1.shvnama = &varname1;
 SHVAR1.shvnaml = sizeof(varname1)-1;
 
 SHVAR1.shvvala = &varvalue1;
 SHVAR1.shvvall = sizeof(varvalue1)-1;

 excomfunc = fetch("IRXEXCOM");
 if (excomfunc==NULL) { perror("fetch IRXEXCOM failed:"); exit(1); }

 fprintf (stderr, "C main beginning %d args: <1=%s>, <__osplist=%x>\n",argc, argv[0], __osplist);

 cnt=0;
 while (ARGTEP[cnt].argtable_argstring_ptr != (void *) -1)
 {
   fprintf (stderr, "arg[%03d]=<%.*s>\n", 
            cnt,
            ARGTEP[cnt].argtable_argstring_length,
            ARGTEP[cnt].argtable_argstring_ptr);
   cnt++;
 }

 #if 0
 for (cnt=0; environ[cnt]!=NULL; cnt++)
    fprintf (stderr, "  environ[%.3d]=%s\n", environ[cnt]);

 {
   #if 0
   char * envvar = getenv("DIV0");
   if (envvar!=NULL)
   #endif
     cnt/=0;
 }
 #endif

 excomfunc("IRXEXCOM", NULL, NULL, SHVAR1, envbp0, retcode);

 return(0);
}





edcpipmz.c
/**********************************************************************/
/* COPYRIGHT IBM CORP. 2012, 2013                                     */
/**********************************************************************/

#include <stdio.h>
#include <stdlib.h>
#include <string.h>
#include <time.h>
#include <errno.h>

#include <irxevalb.h>
#include <irxargtb.h>
#include <irxinstb.h>     

typedef struct irxexec_flags_s {
  unsigned int 
    command : 1, 
    function : 1,
    subroutine : 1 ,
    extended_retcodes : 1 ,
    _reserved : 4;
  char _reseved_flags[3];
} IRXEXEC_FLAGS;

typedef int (IRXEXEC)(void** execblk, void** argtable, IRXEXEC_FLAGS flags, 
                      void** instor, void** cppl, void** evalb,
                      void** work, void** user,
                      void** envb, void* retcd );
#pragma linkage(IRXEXEC, OS)

int aa = 0;
void *envbp0=0, *null=NULL, *userfld1=0;

IRXEXEC_FLAGS execflags = {0, 0, 1, 1, 0};

main (int argc, char **argv)
{


 #define EVDATA_SIZE 240
 struct eval1 {
  struct evalblock EVB1;
  char restofevdata[EVDATA_SIZE];
 } EVAL1 = { 0, ((EVDATA_SIZE/8)+2), 0};  /* how do they get 250, has to be 240 or 248 ? */
 unsigned char *EVAL1data_ptr = &EVAL1.EVB1.evalblock_evdata;

 #define ARG1 "hello"
 #define ARG2 " REXX "
 #define ARG3 "from LE"
 #define ARGEND "\xff\xff\xff\xff\xff\xff\xff\xff"

 struct argtable1 {
   struct argtable_entry args1[3];
   struct argstring args1end;
 } ARGTABLE1 = 
 {ARG1, sizeof(ARG1)-1,
  ARG2, sizeof(ARG2)-1,
  ARG3, sizeof(ARG3)-1,
  ARGEND};

 char *REXX1[] = {
   "Say; Say COPIES('=',72)"
  ,"Parse Arg arg1, arg2, arg3"
  ,"Say \"Args to REXX are:\" arg1\",\" arg2\",\" arg3\".\""
  ,"Parse Source src"
  ,"Say \"REXX source is '\"src\"'\""
  ,"Say COPIES('=',72); Say"
  ,"Exit 72"
 };

 #define REXXIN1_CNT (sizeof(REXX1)/sizeof(char*))
 struct instblk_entry REXXIN1[REXXIN1_CNT];

 struct instblk_header INSTBLK1 = 
 {"IRXINSTB",
  sizeof(INSTBLK1),
  0,
  0, /* address of INSTOR exec */
  sizeof(struct instblk_entry)*REXXIN1_CNT,
  "inSideMe",
  "-ddNone-",
  "SH      ",
  0,
  8,
  "NoDSname",
  0};


 IRXEXEC* execfunc;

 int retcode, RC, rr;

 struct argtable1      *ARGTABLE1_ptr = &ARGTABLE1;
 struct eval1          *EVAL1_ptr     = &EVAL1;
 struct instblk_header *INSTBLK1_ptr  = &INSTBLK1;

 INSTBLK1._instblk_address = &REXXIN1[0];

 for (rr=0; rr<REXXIN1_CNT; rr++) {
   REXXIN1[rr].instblk_stmt_ = REXX1[rr];
   REXXIN1[rr].instblk_stmtlen = strlen(REXX1[rr]);
 }

 execfunc = (IRXEXEC*) fetch("IRXEXEC");
 if (execfunc==NULL) { perror("fetch IRXEXEC failed:"); exit(1); }

 fprintf (stderr, "C main beginning %d args:\n",argc);
 for (aa=0; aa<argc; aa++) {
   printf("arg[%03d]=<%s>\n", aa, argv[aa]);
 }

 ARGTABLE1.args1[0].argtable_argstring_ptr = argv[1];
 ARGTABLE1.args1[0].argtable_argstring_length = strlen(argv[1]);

 RC=execfunc(&null,
             (void**) &ARGTABLE1_ptr,
             execflags,
             (void**)&INSTBLK1_ptr,
             &null,
             (void**)&EVAL1_ptr,
             &null,
             &userfld1,
             &envbp0,
             &retcode);

 printf("retcode is: %d\n", retcode);
 printf("EVAL_data is: %.8s\n", EVAL1data_ptr);

 if (RC==0) RC=strtoul((char *)EVAL1data_ptr,NULL,0);

 exit(RC);

}





igzpipim2.cbl
       CBL LIB,QUOTE
      *
      *****************************************************************
      * COPYRIGHT IBM CORP. 2012, 2013                                *
      *****************************************************************
      *

       IDENTIFICATION DIVISION.
       PROGRAM-ID. COBPIPM2.

       DATA DIVISION.
       WORKING-STORAGE SECTION.
           COPY IRXSHVB.
      *
       77 REXXRTN         PIC X(8) VALUE "IRXEXCOM".
      *77 REXXRTN         PIC X(8) VALUE "SAYARGS ".
      *
       77 REXXRC          PIC S9(9) BINARY.
       77 REXXRTNRC       PIC S9(9) BINARY.
       77 SHVRETB         PIC S9(9) BINARY.
      *
       77 NULADDR         PIC 9(9) USAGE BINARY VALUE 0.
      *77 SHVADDR         USAGE POINTER.
      *
      *01 VarName1.
      *   02 VarName1L    PIC S9(9).
      *   02 VarName1V    PIC X VALUE "LEREXX"
       01 VarName1  VALUE "006LEREXX".
          02 VarName1L    PIC 999.
          02 VarName1V    PIC X
             Occurs 1 To 250 Times Depending on VarName1L.
      *
       01 VarValue1 VALUE "023and COBOL all together!".
          02 VarValue1L   PIC 999.
          02 VarValue1V   PIC X
             Occurs 1 To 250 Times Depending on VarValue1L.
      *
      *************************************************
       PROCEDURE DIVISION.
      *************************************************
      *************************************************
       PARA-REXXVAR.
           Initialize SHVBLOCK Replacing AlphaNumeric By X'00'.
      * STORE (direct not symbolic) the variable
           Move SHVSTORE To SHVCODE Of SHVBLOCK.
      * The variable name
           Set SHVNAMA Of SHVBLOCK To Address Of VARNAME1V(1).
           Move VARNAME1L To SHVNAML Of SHVBLOCK.
      * The variable value
           Set SHVVALA Of SHVBLOCK To Address Of VARVALUE1V(1).
           Move VARVALUE1L To SHVVALL Of SHVBLOCK.
      * Call the REXX service as a subprogram to set the shared variable
           CALL REXXRTN USING REXXRTN ,
                OMITTED , OMITTED ,
                SHVBLOCK ,
                NULADDR , REXXRC
            Returning REXXRTNRC.
      *
           MOVE SHVRET To SHVRETB.
           DISPLAY "REXX RC IS " REXXRC ","
                  " IRXEXCOM RC IS " REXXRTNRC ","
                  " SHVRET IS " SHVRETB "." .
      *
           GOBACK.
      *




IRXSHVB.cbl
      *****************************************************************
      * COPYRIGHT IBM CORP. 2012, 2013                                *
      *****************************************************************

      *****************************************************************
      *****************************************************************
      **  Shared REXX Variable Request Block mapping.                **
      *****************************************************************
      *****************************************************************

      *****************************************************************
      *   SHARED VARIABLE REQUEST BLOCK                               *
      *****************************************************************
       01  SHVBLOCK.
      *       Chain pointer to next SHVBLOCK
           02  SHVNEXT             USAGE POINTER.
      *       Used during "FETCH NEXT"
      *        Contains length of buffer
      *        pointed to by SHVNAMA
           02  SHVUSER             PIC S9(9) BINARY.
           02  SHVCODES.
      *       Function code - indicates type
               03 SHVCODE          PIC A.
      *       Return codes
               03 SHVRET           PIC X.
      *       Reserved (should be 0)
           02                      PIC X(2).
      *       Length of fetch value buffer
           02  SHVBUFL             PIC S9(9) BINARY.
      *
      *       Address of variable name
           02  SHVNAMA             USAGE POINTER.
      *       Length of variable name
      *        (Set on fetch)
           02  SHVNAML             PIC S9(9) BINARY .
      *       Address of value buffer
           02  SHVVALA             USAGE POINTER.
      *       Length of value buffer
           02  SHVVALL             PIC S9(9) BINARY .
      *
      *    Length of SHVBLOCK
       77   SHVBLEN                PIC 9(9) BINARY VALUE IS 32 .
      *
      *****************************************************************
      *   SHARED VARIABLE REQUEST BLOCK - function codes              *
      *****************************************************************
      *    Set variable from given value
       77   SHVSTORE               PIC A VALUE IS 'S'.
      *    Copy value of variable to Buffer
       77   SHVFETCH               PIC A VALUE IS 'F'.
      *    Drop variable
       77   SHVDROPV               PIC A VALUE IS 'D'.
      *    Symbolic name Set variable
       77   SHVSYSET               PIC A VALUE IS 's'.
      *    Symbolic name Fetch variable
       77   SHVSYFET               PIC A VALUE IS 'f'.
      *    Symbolic name DROP variable
       77   SHVSYDRO               PIC A VALUE IS 'd'.
      *    Fetch next variable
       77   SHVNEXTV               PIC A VALUE IS 'N'.
      *    Fetch private information
       77   SHVPRIV                PIC A VALUE IS 'P'.

      *****************************************************************
      *   SHARED VARIABLE REQUEST BLOCK - return codes (SHVRET)       *
      *****************************************************************
      *   Execution was OK
       77   SHVCLEAN               PIC X VALUE IS X'00'.
      * Variable did not exist
       77   SHVNEWV                PIC X VALUE IS X'01'.
      * Last variable transferred ("N")
       77   SHVLVAR                PIC X VALUE IS X'02'.
      * Truncation occurred for "Fetch"
       77   SHVTRUNC               PIC X VALUE IS X'04'.
      * Invalid variable name
       77   SHVBADN                PIC X VALUE IS X'08'.
      * Invalid value specified
       77   SHVBADV                PIC X VALUE IS X'10'.
      * Invalid function code (SHVCODE)
       77   SHVBADF                PIC X VALUE IS X'80'.




subr1.rexx
/* REXX */
/**********************************************************************/
/* COPYRIGHT IBM CORP. 2011, 2013                                     */
/**********************************************************************/

Numeric Digits 10 /* needed for STinitD (an address) */

ans = SUBR1()
Say 'EVdata size returned='LENGTH(ans) 

evsize = SUBSTR(ans,1,4); Say 'EVsize (length available without calling IRXRLT(GETBLOCK))='C2D(evsize)*8-16

SThdr=SUBSTR(ans,5,20) /* SUBCOMTB */
STadrD = C2D(SUBSTR(SThdr,1,4))
STtotD = C2D(SUBSTR(SThdr,5,4))
STusedD= C2D(SUBSTR(SThdr,9,4))
STelenD= C2D(SUBSTR(SThdr,13,4))
STinitD= C2D(SUBSTR(SThdr,17,4))

Say 'SUBCOM Table:'
Say ' total entries:' STtotD
Say ' used  entries:' STusedD
Say ' initial environment:' STORAGE(D2X(STinitD),8)
Say ' all environments (token length='STelenD-16'):'
Do ss=1 To STusedD
  STentry = STORAGE(D2X(STadrD),STelenD)
  Say '  'RIGHT(ss,5)' name='SUBSTR(STentry,1,8) 'routine='SUBSTR(STentry,9,8) 'token='SUBSTR(STentry,17)'.'
  STadrD = STadrD + STelenD
End ss

Exit 0




igzpipi2.cbl
       CBL LIB,QUOTE
      *
      *****************************************************************
      * COPYRIGHT IBM CORP. 2012, 2013                                *
      *****************************************************************
      *

       IDENTIFICATION DIVISION.
       PROGRAM-ID. HLLPIPI.

       DATA DIVISION.
       WORKING-STORAGE SECTION.
           COPY IRXSHVB.
      *
       77 CNT             PIC S9(9) BINARY VALUE 0.
      *
      *************************************************
       PROCEDURE DIVISION.
      *************************************************
      *************************************************
       PARA-HLLPIPI.
      *
           Compute CNT = CNT + 1.
           If CNT = 1 Then 
             Display "HLLPIPI COBOL subprogram, first call, using LE PIP
      -              "I Call Sub interface." 
           Else
             Display "HLLPIPI COBOL subprogram called " CNT " times."
           End-If
      *
           Display "HLLPIPI COBOL subprogram returning."
           GoBack.
      *



